Application of J-integral to seismic pressures on frictionless rigid retaining walls

This notebook details the calculations presented in “An application of path-independent integrals to seismic pressures on frictionless rigid retaining walls”, and shows how the plots presented there are generated.

Exact solution

This first section retrieves the implementation of the exact solution as derived in Garcia-Suarez, Joaquin, and Domniki Asimaki. “Exact seismic response of smooth rigid retaining walls resting on stiff soil.” *International Journal for Numerical and Analytical Methods in Geomechanics* 44.13 (2020): 1750-1769..

Auxiliary Parameters

Some basic parameters that appear, over and over, in the solution: respectively, the dimensionless frequencies r and r_c=r/c (both including the damping), the dimensionless wavenumber of S waves α and of P waves β.

In[1]:=
```
c[ν_] = Sqrt[(2 (1 - ν))/(1 - 2 ν)];
rd[r_, δ_] = r/Sqrt[1 + ⅈ * δ];
rc[r_, δ_, ν_] = (r / c[ν])/Sqrt[1 + ⅈ * δ];
α[k_, r_, δ_, ν_] = Sqrt[k^2 - rd[r, δ]^2];
β[k_, r_, δ_, ν_] = Sqrt[k^2 - rc[r, δ, ν]^2];
```

Coefficients

The solution of the system that yields the thrust, along with all the kinematic variables, depends on four constants (Ac, Bc, Cc, Dc) that still depend on the horizontal wavelength and the frequency.

In[6]:=
```
Ac[k_, r_, δ_, v_] =
  ((-2 + c[v]^2) (2 + ⅇ^α[k,r,δ,v]) (-1 + ⅇ^β[k,r,δ,v])^2 k^4 - c[v]^2 ⅇ^α[k,r,δ,v] (1 + ⅇ^(2 β[k,r,δ,v])) α[k, r, δ, v]^2
      β[k, r, δ, v]^2 + k^2 (-2 (-2 + c[v]^2) ⅇ^(α[k,r,δ,v]+β[k,r,δ,v]) α[k, r, δ, v]^2 - 2 (-2 + c[v]^2)
         α[k, r, δ, v] × β[k, r, δ, v] + 2 (-2 + c[v]^2) ⅇ^(2 β[k,r,δ,v]) α[k, r, δ, v] × β[k, r, δ, v] +
        4 c[v]^2 ⅇ^β[k,r,δ,v] β[k, r, δ, v]^2 + ⅇ^α[k,r,δ,v] ((-2 + c[v]^2) α[k, r, δ, v]^2 -
           4 α[k, r, δ, v] × β[k, r, δ, v] - c[v]^2 β[k, r, δ, v]^2) + ⅇ^(α[k,r,δ,v]+2 β[k,r,δ,v])
         ((-2 + c[v]^2) α[k, r, δ, v]^2 + 4 α[k, r, δ, v] × β[k, r, δ, v] - c[v]^2 β[k, r, δ, v]^2))) /
   (c[v]^2 Sqrt[2 * π] × β[k, r, δ, v] ((1 + ⅇ^(2 β[k,r,δ,v])) α[k, r, δ, v] × β[k, r, δ, v]
        (-(-6 + c[v]^2) k^4 - (-2 + c[v]^2) k^2 α[k, r, δ, v]^2 + c[v]^2 (k^2 + α[k, r, δ, v]^2) β[k, r, δ, v]^2)
        Cosh[α[k, r, δ, v]] + 2 ⅇ^β[k,r,δ,v] k^2
        (-2 α[k, r, δ, v] × β[k, r, δ, v] (-(-3 + c[v]^2) k^2 + α[k, r, δ, v]^2 + c[v]^2 β[k, r, δ, v]^2) +
          ((-2 + c[v]^2) k^2 (k^2 + α[k, r, δ, v]^2) - (c[v]^2 k^2 + (4 + c[v]^2) α[k, r, δ, v]^2) β[k, r, δ, v]^2)
           Sinh[α[k, r, δ, v]] × Sinh[β[k, r, δ, v]])));
```

In[7]:=
```
Bc[k_, r_, δ_, v_] =
  (ⅇ^(-α[k,r,δ,v]-β[k,r,δ,v]) (-(-2 + c[v]^2) (1 + 2 ⅇ^α[k,r,δ,v]) (-1 + ⅇ^β[k,r,δ,v])^2 k^4 + c[v]^2 (1 + ⅇ^(2 β[k,r,δ,v]))
        α[k, r, δ, v]^2 β[k, r, δ, v]^2 + k^2 (-(-2 + c[v]^2) (-1 + ⅇ^β[k,r,δ,v])^2 α[k, r, δ, v]^2 +
          2 (2 + (-2 + c[v]^2) ⅇ^α[k,r,δ,v]) (-1 + ⅇ^(2 β[k,r,δ,v])) α[k, r, δ, v] × β[k, r, δ, v] +
          c[v]^2 (1 + ⅇ^(2 β[k,r,δ,v]) - 4 ⅇ^(α[k,r,δ,v]+β[k,r,δ,v])) β[k, r, δ, v]^2))) / (2 c[v]^2 Sqrt[2 π] β[k, r, δ, v]
     (α[k, r, δ, v] × β[k, r, δ, v] (-(-6 + c[v]^2) k^4 - (-2 + c[v]^2) k^2 α[k, r, δ, v]^2 +
          c[v]^2 (k^2 + α[k, r, δ, v]^2) β[k, r, δ, v]^2) Cosh[α[k, r, δ, v]] × Cosh[β[k, r, δ, v]] +
       k^2 (-2 α[k, r, δ, v] × β[k, r, δ, v] (-(-3 + c[v]^2) k^2 + α[k, r, δ, v]^2 + c[v]^2 β[k, r, δ, v]^2) +
          ((-2 + c[v]^2) k^2 (k^2 + α[k, r, δ, v]^2) - (c[v]^2 k^2 + (4 + c[v]^2) α[k, r, δ, v]^2) β[k, r, δ, v]^2)
           Sinh[α[k, r, δ, v]] × Sinh[β[k, r, δ, v]])));
```

In[8]:=
```
Cc[k_, r_, δ_, v_] =
  (2 α[k, r, δ, v] ((1 - (-2 + c[v]^2) ⅇ^β[k,r,δ,v]) k^2 + α[k, r, δ, v]^2) * β[k, r, δ, v] + α[k, r, δ, v]
      (-4 ⅇ^β[k,r,δ,v] k^2 + (-2 + c[v]^2) (k^2 + α[k, r, δ, v]^2)) β[k, r, δ, v] × Cosh[α[k, r, δ, v]] -
     (k^2 + α[k, r, δ, v]^2) ((-2 + c[v]^2) (-1 + ⅇ^β[k,r,δ,v]) k^2 - c[v]^2 ⅇ^β[k,r,δ,v] β[k, r, δ, v]^2)
      Sinh[α[k, r, δ, v]]) / (c[v]^2 Sqrt[2 π] × β[k, r, δ, v]
     (α[k, r, δ, v] × β[k, r, δ, v] (-(-6 + c[v]^2) k^4 - (-2 + c[v]^2) k^2 α[k, r, δ, v]^2 +
          c[v]^2 (k^2 + α[k, r, δ, v]^2) β[k, r, δ, v]^2) Cosh[α[k, r, δ, v]] × Cosh[β[k, r, δ, v]] +
       k^2 (-2 α[k, r, δ, v] × β[k, r, δ, v] (-(-3 + c[v]^2) k^2 + α[k, r, δ, v]^2 + c[v]^2 β[k, r, δ, v]^2) +
          ((-2 + c[v]^2) k^2 (k^2 + α[k, r, δ, v]^2) - (c[v]^2 k^2 + (4 + c[v]^2) α[k, r, δ, v]^2) β[k, r, δ, v]^2)
           Sinh[α[k, r, δ, v]] × Sinh[β[k, r, δ, v]])));
```

In[9]:=
```
Dc[k_, r_, δ_, v_] =
  (ⅇ^(-α[k, r, δ, v]-β[k, r, δ, v]) ((-2 + c[v]^2) (-1 + ⅇ^(2 α[k, r, δ, v])) (-1 + ⅇ^(β[k, r, δ, v])) k^4 + α[k, r, δ, v]^2
        β[k, r, δ, v] (c[v]^2 (-1 + ⅇ^(2 α[k, r, δ, v])) β[k, r, δ, v] - 2 ⅇ^(α[k, r, δ, v]+β[k, r, δ, v])
          α[k, r, δ, v] (2 + (-2 + c[v]^2) Cosh[α[k, r, δ, v]])) +
       k^2 ((-2 + c[v]^2) (-1 + ⅇ^(2 α[k, r, δ, v])) (-1 + ⅇ^(β[k, r, δ, v])) α[k, r, δ, v]^2 + 4 α[k, r, δ, v] ×
          β[k, r, δ, v] + c[v]^2 (-1 + ⅇ^(2 α[k, r, δ, v])) β[k, r, δ, v]^2 - 2 ⅇ^(α[k, r, δ, v]) α[k, r, δ, v] × β[k,
            r, δ, v] (-2 (-2 + c[v]^2 + ⅇ^(α[k, r, δ, v])) + ⅇ^(β[k, r, δ, v]) (2 + (-2 + c[v]^2) Cosh[α[k, r, δ, v]]))))) /
   (2 c[v]^2 * Sqrt[2 π] * β[k, r, δ, v] (α[k, r, δ, v] × β[k, r, δ, v]
        (-(-6 + c[v]^2) k^4 - (-2 + c[v]^2) k^2 α[k, r, δ, v]^2 + c[v]^2 (k^2 + α[k, r, δ, v]^2) β[k, r, δ, v]^2)
        Cosh[α[k, r, δ, v]] × Cosh[β[k, r, δ, v]] +
       k^2 (-2 α[k, r, δ, v] × β[k, r, δ, v] (-(-3 + c[v]^2) k^2 + α[k, r, δ, v]^2 + c[v]^2 β[k, r, δ, v]^2) +
          ((-2 + c[v]^2) k^2 (k^2 + α[k, r, δ, v]^2) - (c[v]^2 k^2 + (4 + c[v]^2) α[k, r, δ, v]^2) β[k, r, δ, v]^2)
           Sinh[α[k, r, δ, v]] × Sinh[β[k, r, δ, v]])));
```

Variables

Thrust(s)

Next, we introduce the exact thrust and the one presented by Veletsos and Younan.

Exact solution

In[10]:=
```
Q[r_, δ_, v_] :=
  2/Sqrt[2 π] NIntegrate[Sqrt[2/π]/β[k, r, δ, v]^2 + Ac[k, r, δ, v] * (c[v]^2 - 2 * Exp[-α[k, r, δ, v]]) +
     Bc[k, r, δ, v] * (c[v]^2 - 2 * Exp[α[k, r, δ, v]]) + (c[v]^2 (1 - (k/β[k, r, δ, v])^2) - 2) *
      (Cc[k, r, δ, v] * Exp[-β[k, r, δ, v]] + Dc[k, r, δ, v] * Exp[β[k, r, δ, v]]) +
     ((c[v] * k)/β[k, r, δ, v])^2 * (Cc[k, r, δ, v] + Dc[k, r, δ, v]), {k, 0, 10}];
```

Veletsos-Younan

In[11]:=
```
QVY[r_, δ_, v_] := (32/π^3)/Sqrt[(1 - v) (2 - v)] Sum[1/(2 n - 1)^3 1/Sqrt[1 - ((r/(π/2 (2 n - 1)))/Sqrt[1 + ⅈ * δ])^2], {n, 1, 5}]
```

Comparison

A quick comparison between Younan-Veletsos expression and the exact solution, for Poisson's ratio v=1/3 and damping factor δ_d=1% (recall that all expressions are dimensionless, see, e.g., the eq.(13) in the manuscript and its appendix for details).

In[12]:=
```
NU = 1/3; DE = 0.01;
```

We use parallel functions to evaluate the exact solution numerically more efficiently.

In[13]:=
```
Nkernels = 8; (*# of kernels available*)
CloseKernels[];
LaunchKernels[Nkernels];
```

In[16]:=
```
rlist = Table[ii/100., {ii, 10, Ceiling[350. * π/2]}];
```

In[17]:=
```
exactQ = ParallelTable[{rlist[[jj]], Abs[Quiet[Q[rlist[[jj]], DE, NU]]]},
    {jj, 1, Length[rlist]}]; // AbsoluteTiming
```

Out[17]=
```
{9.96379, Null}
```

In[18]:=
```
fig1 = ListLinePlot[exactQ, PlotRange → {{0, 3.5 * π / 2}, {0, 10}},
   Ticks → {{0, Pi/2, Pi, 3 Pi/2}, {0, 1, 5, 10}}, PlotStyle → Orange,
   PlotLegends → {Style["Exact", FontColor → LightGray]}];
fig2 = Plot[{Abs[QVY[r, DE, NU]]}, {r, 0.1, 3.5 * π / 2},
   PlotRange → {{0, 3.5 * π / 2}, {0, 10}}, Ticks → {{0, Pi/2, Pi, 3 Pi/2}, {0, 1, 5, 10}},
   PlotLegends → {Style["Younan-Veletsos", FontColor → LightGray]}];
Show[fig2, fig1]
```

Out[19]=



See how the peak in the exact solution at r=π is not captured by the model.

Simplified model enabled by the J-integral

The following creates the terms necessary to evaluate the thrust estimate provided by the new model.

Auxiliary parameters

The quasi-static value of κ, whose derivation is much easier (see section D.43), is also included for verification purposes:

In[20]:= $$\texttt{κst}[v\_] := \frac{\texttt{c}[v]}{\texttt{Sqrt}\left[(2 * \texttt{c}[v]^2 - 1) + \frac{672 * \texttt{Catalan} * \texttt{Zeta}[3]}{\pi^5} (\texttt{c}[v]^2 - 1) - \frac{192 * \texttt{Catalan}^2}{\pi^4}\right]}$$

It follows finally the term k_n (that serves to consider different modes):

In[21]:= $$\texttt{kn}[N\_] := \frac{\pi}{2} (2\,N - 1)$$

Definition of coefficients

Find next the coefficients that appear in the expression of κ:

In[22]:= $$\texttt{un}[\delta\_, r\_, N\_] := -\frac{2}{\texttt{kn}[N]} \frac{1}{(\texttt{kn}[N]^2 - \texttt{rd}[r, \delta]^2)}$$

In[23]:= $$\texttt{CN}[v\_, \delta\_, r\_, N\_] := -\frac{(\texttt{c}[v]^2 - 1)}{\texttt{c}[v]^2} \frac{2}{(\texttt{kn}[N]^2 - \texttt{rc}[r, \delta, v]^2)\, \texttt{Sqrt}\left[\texttt{kn}[N]^2 - \texttt{rd}[r, \delta]^2\right]}$$

In[24]:= $$\texttt{AN}[v\_, \delta\_, r\_, N\_] := -\texttt{Sum}[\texttt{CN}[v, \delta, r, n], \{n, 1, N\}]$$

In[25]:= $$\texttt{BN}[v\_, \delta\_, r\_, N\_] := \texttt{AN}[v, \delta, r, N] * \texttt{Tan}[\texttt{rc}[r, \delta, v]] + \texttt{Sum}\left[\frac{\texttt{CN}[v, \delta, r, n] * \texttt{kn}[n] * (-1)^{n+1}}{\texttt{rc}[r, \delta, v] * \texttt{Cos}[\texttt{rc}[r, \delta, v]]} \left(1 - \left(\frac{\texttt{c}[v]^2 - 2}{\texttt{c}[v]^2 - 1}\right) * \frac{\texttt{kn}[n]^2 - \texttt{rc}[r, \delta, v]^2}{\texttt{kn}[n]^2}\right), \{n, 1, N\}\right]$$

Definition of κ

A preliminary note: all the names ending in “N” denote a variable that has to be evaluated as a combination of N modes, where N is the number of modes the user choses (in this case, N=2 was used and shown to be enough).

First, consider the right-hand side (RHS) of the J-integral:

In[26]:= $$\texttt{RHSN}[v\_, \delta\_, r\_, N\_] := \texttt{Sum}\left[\frac{1}{\texttt{kn}[n]^2} \left(\frac{1}{\texttt{kn}[n]^2 - \texttt{rd}[r, \delta]^2}\right), \{n, 1, N\}\right]$$

Then, consider the left-hand (LHS) side of the same equation, which takes a more involved form due to the displacement field that the model yields. Therefore, an auxiliary function (Aux1) is introduced. Note that the variable LHSN actually is the left-hand side divided by κ^2, that is why later we can divide left over right later to obtain κ.

In[27]:=
```
Aux1N[ν_, δ_, r_, N_] := -Sin[2*rc[r, δ, ν]]/2 (BN[ν, δ, r, N]^2 - AN[ν, δ, r, N]^2) +
  2*BN[ν, δ, r, N]*Sum[CN[ν, δ, r, n], {n, 1, N}] +
  2*AN[ν, δ, r, N]*BN[ν, δ, r, N]*Sin[rc[r, δ, ν]]^2
```

In[28]:=
```
LHSN[ν_, δ_, r_, N_] :=
 Sum[c[ν]^2/4 (kn[n]^2 - rd[r, δ]^2)*un[δ, r, n]^2 + 1/2 (-(c[ν]^2 kn[n]^2 - rd[r, δ]^2)/2 CN[ν, δ, r, n]^2),
   {n, 1, N}] + 1/2 (c[ν]*rd[r, δ]*Aux1N[ν, δ, r, N])
```

In[29]:=
```
κdN[ν_, δ_, r_, N_] := Sqrt[RHSN[ν, δ, r, N]/LHSN[ν, δ, r, N]]
```

Compare the dynamic κ to the static κ

Next figure corresponds to Figure B.8 in the appendix. See that, in the following plots, the variable displayed in the horizontal axis is r, not $\frac{\omega}{\omega_s} = \frac{2}{\pi} r$, as in the actual figure that can be found in the text (to go from one to the other just a trivial re-scaling of the horizontal axis is necessary).

Assign some values to the parameters:

In[30]:=
```
DE = 0.05; NU = 0.1;
```

In[31]:=
```
Plot[{Abs[κdN[NU, DE, r, 2]], κst[NU]}, {r, 0, 3.5 π}, PlotRange → {{0, 5/4 π}, {0, 1}},
 Ticks → {{0, Pi/2, Pi, 3 Pi/2}, {0, 0.5, 1}},
 PlotStyle → {Red, Yellow},
 PlotLegends → "Expressions"]
```

Out[31]=



The dip in the dynamic κ corrects the dynamic amplitude to better match the exact solution.

Integral of gradient of horizontal displacement at the wall

In[32]:=
```
intdudxstN[δ_, r_, N_] := Sum[Sqrt[kn[n]^2 - rd[r, δ]^2]/kn[n] * un[δ, r, n], {n, 1, N}]
```

Vertical displacement at the top of the wall

In[33]:=
```
vdynN[ν_, δ_, r_, η_, N_] := Sum[CN[ν, δ, r, n] * Cos[π/2 (2 n - 1) η], {n, 1, N}] +
  AN[ν, δ, r, N] * Cos[rc[r, δ, ν] * η] + BN[ν, δ, r, N] * Sin[rc[r, δ, ν] * η]
```

Thrust estimate

Just combining properly these two pieces:

In[34]:=
```
Qest[ν_, δ_, r_, N_] := c[ν]^2 * intdudxstN[δ, r, N] + (c[ν]^2 - 2) * vdynN[ν, δ, r, 1, N]
```

Plots

Figure 3(a)

In[35]:=
```
NU = 1 / 10; DE = 0.05;
```

In[36]:=
```
exactQ = ParallelTable [{rlist[[jj]], Abs[Quiet[Q[rlist[[jj]], DE, NU]]]},
    {jj, 1, Length[rlist]}]; // AbsoluteTiming
```

Out[36]=
```
{10.0398 , Null}
```

In[37]:=
```
xticks = {0, {Pi/2, 1}, {Pi, 2}, {3 Pi/2, 3}};
yticks = Range[5];
```

In[39]:=
```
fig1 = ListLinePlot[exactQ, PlotRange → {{0, 3.5 * π / 2}, {0, 5}}, Ticks → {xticks, yticks},
  PlotStyle → Red,
  PlotLegends → Placed[{Style["Exact", FontColor → LightGray]}, Scaled[{0.71, 0.75}]]];
fig2 =
 Plot[{Abs[QVY[r, DE, NU]], Abs[κdN[NU, DE, r, 2] * Qest[NU, DE, r, 2]]}, {r, 0.1, 3.5 * π / 2},
  PlotRange → {{0, 3.5 * π / 2}, {0, 5}},
  Ticks → {xticks, yticks},
  PlotStyle → {Blue, LightGray},
  PlotLegends → Placed[{Style["Younan-Veletsos",
      FontColor → LightGray],
     Style["using κ",
      FontColor → LightGray]}, Scaled[{0.8, 0.6}]]
 ];
fig3a = Show[fig2, fig1];
```

Figure 3(b)

In[41]:=
```
NU = 1 / 3;
```

In[42]:=
```
exactQ = ParallelTable[{rlist[[jj]], Abs[Quiet[Q[rlist[[jj]], DE, NU]]]},
    {jj, 1, Length[rlist]}]; // AbsoluteTiming
```

Out[42]=
```
{10.2115, Null}
```

In[43]:=
```
fig1 = ListLinePlot[exactQ, PlotRange → {{0, 3.5 * π / 2}, {0, 5}},
  Ticks → {xticks, yticks},
  PlotStyle → Red];
fig2 =
 Plot[{Abs[QVY[r, DE, NU]], Abs[κdN[NU, DE, r, 2] * Qest[NU, DE, r, 2]]}, {r, 0.1, 3.5 * π / 2},
  PlotRange → {{0, 3.5 * π / 2}, {0, 5}},
  Ticks → {xticks, yticks},
  PlotStyle → {Blue, LightGray}];
fig3b = Show[fig2, fig1];
```

Figure 3(c)

In[45]:=
```
NU = 7 / 16;
```

In[46]:=
```
exactQ = ParallelTable[{rlist[[jj]], Abs[Quiet[Q[rlist[[jj]], DE, NU]]]},
    {jj, 1, Length[rlist]}]; // AbsoluteTiming
```

Out[46]=
```
{10.272, Null}
```

In[47]:=
```
fig1 = ListLinePlot[exactQ, PlotRange → {{0, 3.5 * π / 2}, {0, 5}},
   Ticks → {xticks, yticks},
   PlotStyle → Red];
fig2 =
  Plot[{Abs[QVY[r, DE, NU]], Abs[κdN[NU, DE, r, 2] * Qest[NU, DE, r, 2]]}, {r, 0.1, 3.5 * π / 2},
   PlotRange → {{0, 3.5 * π / 2}, {0, 5}},
   Ticks → {xticks, yticks},
   PlotStyle → {Blue, LightGray}];
fig3c = Show[fig2, fig1];
```

Figure 3

In[49]:=
```
GraphicsGrid[{{fig3a}, {fig3b}, {fig3c}}, ImageSize → Large]
```

Out[49]=



Figure 4(b)

In[50]:=
```
DE = 0.1;
```

In[51]:=
```
exactQ = ParallelTable [{rlist[[jj]], Abs[Quiet[Q[rlist[[jj]], DE, NU]]]},
    {jj, 1, Length[rlist]}]; // AbsoluteTiming
```

Out[51]=
```
{10.5706, Null}
```

In[52]:=
```
fig1 = ListLinePlot [exactQ, PlotRange → {{0, 3.5 * π / 2}, {0, 5}}, Ticks → {xticks, yticks},
  PlotStyle → Red,
  PlotLegends → Placed[{Style["Exact", FontColor → LightGray]}, Scaled[{0.71, 0.75}]]];
fig2 =
 Plot[{Abs[QVY[r, DE, NU]], Abs[κdN[NU, DE, r, 2] * Qest[NU, DE, r, 2]]}, {r, 0.1, 3.5 * π / 2},
  PlotRange → {{0, 3.5 * π / 2}, {0, 5}},
  Ticks → {xticks, yticks},
  PlotStyle → {Blue, LightGray},
  PlotLegends → Placed[{Style["Younan-Veletsos",
      FontColor → LightGray],
     Style["using κ",
      FontColor → LightGray]}, Scaled[{0.8, 0.6}]]
 ];
fig4b = Show[fig2, fig1];
```

Figure 4(c)

In[54]:=
```
DE = 0.15;
```

In[55]:=
```
exactQ = ParallelTable [{rlist[[jj]], Abs[Quiet[Q[rlist[[jj]], DE, NU]]]},
    {jj, 1, Length[rlist]}]; // AbsoluteTiming
```

Out[55]=
```
{10.7529, Null}
```

In[56]:=
```
fig1 = ListLinePlot [exactQ, PlotRange → {{0, 3.5 * π / 2}, {0, 5}}, Ticks → {xticks, yticks},
  PlotStyle → Red];
fig2 =
 Plot[{Abs[QVY[r, DE, NU]], Abs[κdN[NU, DE, r, 2] * Qest[NU, DE, r, 2]]}, {r, 0.1, 3.5 * π / 2},
  PlotRange → {{0, 3.5 * π / 2}, {0, 5}},
  Ticks → {xticks, yticks},
  PlotStyle → {Blue, LightGray}
 ];
fig4c = Show[fig2, fig1];
```

In[58]:=
```
GraphicsGrid[{{fig3c}, {fig4b}, {fig4c}}, ImageSize → Large]
```

Out[58]=



Figure 5

In[59]:=
```
nus = Table[(ii - 1)/200., {ii, 1, 100}];
```

In[60]:=
```
exactQ = ParallelTable [{nus[[jj]], Abs[Quiet[Q[0.1, 0.1, nus[[jj]]]]]}, {jj, 1, Length[nus]}]; //
 AbsoluteTiming
```

Out[60]=
```
{2.36026, Null}
```

In[61]:=
```
fig1 = ListPlot[exactQ,
   PlotRange → {{0, 0.5}, {0, 1.5}},
   Joined → True,
   PlotStyle → Red,
   PlotLegends → Placed[{Style["Exact", FontColor → LightGray]}, Scaled[{0.5, 0.35}]]];
```

In[62]:=
```
fig2 = Plot[{Abs[QVY[0.1, 0.1, x]], Abs[κst[x] * Qest[x, 0.1, 0.1, 2]]}, {x, 0.01, 0.49},
   PlotRange → {{0, 0.5}, {0, 1.5}},
   PlotStyle → {Blue, LightGray},
   PlotLegends → Placed[{
      Style["Veletsos-Younan", FontColor → LightGray],
      Style["using κ", FontColor → LightGray]}
     , Scaled[{0.6, 0.2}]]
  ];
Show[fig1, fig2]
```

Out[63]=



Verifying some integrals

These are some simple integrals that appear in the appendix during the computation of norms of the approximate displacement field from the reduced model (defined up to κ). The following ones correspond to $r^2 \| v \|^2 - c^2 \| \frac{dv}{d\eta} \|^2$:

The one of $C_n{}^2$ -> equation (B.30a)

In[64]:=
```
FullSimplify[Integrate[r^2 * Cos[k * η]^2 - c^2 * k^2 * Sin[k * η]^2, {η, 0, 1}]]
```

Out[64]= $\frac{r^2 (k + \text{Cos}[k] \times \text{Sin}[k])}{2\,k} + \frac{1}{4} c^2\, k\, (-2\,k + \text{Sin}[2\,k])$

See that $\text{Sin}[k] = \text{Sin}[\pi (2\,n - 1)/2] = (-1)^{n+1}$ while Cos[k]=Sin[2k]=0, what yields $\frac{r^2}{2} - \frac{c^2 k^2}{2}$

The one of A^2 (equal minus the one of B^2) -> equation (B.30b)

In[65]:=
```
FullSimplify[Integrate[r^2 * Cos[r/c * η]^2 - r^2 * Sin[r/c * η]^2, {η, 0, 1}]]
```

Out[65]= $\frac{1}{2} c\, r\, \text{Sin}\left[\frac{2\,r}{c}\right]$

The one of 2 AC_n -> equation (B.30c)

In[66]:=
```
FullSimplify[Integrate[r^2 * Cos[k * η] * Cos[r/c * η] - c * r * k * Sin[k * η] * Sin[r/c * η], {η, 0, 1}]]
```

Out[66]= $c\, r\, \text{Cos}[k] \times \text{Sin}\left[\frac{r}{c}\right]$

Since Cos[k]=0, then the whole integral vanishes.

The one of 2 BC_n -> equation (B.30d)

In[67]:=
```
FullSimplify[Integrate[r^2 * Cos[k * η] * Sin[r/c * η] + c * r * k * Sin[k * η] * Cos[r/c * η], {η, 0, 1}]]
```

Out[67]= $c\, r \left(1 - \text{Cos}[k] \times \text{Cos}\left[\frac{r}{c}\right]\right)$

Since Cos[k]=0, then c*r.

The one of 2 BA -> equation (B.30e)

In[68]:=
```
2 * FullSimplify[Integrate[r^2 * Cos[r/c * η] * Sin[r/c * η], {η, 0, 1}]]
```

Out[68]= $c\, r\, \text{Sin}\left[\frac{r}{c}\right]^2$

Verifying some sums

Sum in (B.38)

Note command `/. {k -> π/2 (2 n - 1)}` implies substituting k by that expression before performing the sum

In[69]:=
```
Sum[-1/(k^2 (k^2 - r^2)) /. {k -> π/2 (2 n - 1)}, {n, 1, ∞}]
```

Out[69]=
$$\frac{r - \mathrm{Tan}[r]}{2\, r^3}$$

Sum in (B.28)

In[70]:=
```
Simplify@Sum[(-(c^2 - 1)/c^2 1/((k^2 - (r/c)^2) Sqrt[k^2 - r^2]))^2 /. {k -> π/2 (2 n - 1)}, {n, 1, ∞}]
```

Out[70]=
$$\frac{-\left(\left(-1 + c^2\right) r\, \mathrm{Sec}\left[\frac{r}{c}\right]^2\right) + 2\, \mathrm{Tan}[r] + c\left(-3 + c^2\right) \mathrm{Tan}\left[\frac{r}{c}\right]}{4\, r^5}$$